MaN 15

SC-M-69-335

AVINT: APPROXIMATE INTEGRATOR OF FUNCTIONS TABULATED AT ARBITRARILY SPACED ABSCISSAS

R. E. Jones, 9422
Sandia Laboratories, Albuquerque

July 1969

ABSTRACT

AVINT is a computer subroutine for integration of data tabulated at arbitrarily spaced abscissas.

b

FOREWORD

The Sandia Laboratories Mathematical Program Library consists of a number of dependable, high-quality, general-purpose mathematical computing routines. The standards established for the library require that these routines be mathematically sound, effectively implemented, extensively tested, and thoroughly documented. This report documents one such routine.

The library emphasizes effective coverage of various distinct mathematical areas with a minimum number of routines. Nevertheless, it may contain other routines similar in nature but complementary to the one described here. Additional information on Sandia's mathematical program library, a description of the standard format for documenting these routines, and a guide to other routines in the library are contained in SC-M-69-337.

This report is also identified within Sandia Laboratories as Computing Publication ML0003/ALL. The routine was originally documented in October 1968. This report and its corresponding library routine are expected to be available from COSMIC shortly after publication.

CONTENTS

AVINT: APPROXIMATE INTEGRATOR OF FUNCTIONS TABULATED AT ARBITRARILY SPACED ABSCISSAS

1. Introduction

1.1 Background

AVINT appears in an appendix of FORTRAN programs in Davis and Rabinowitz.[1] The version of AVINT in Sandia's mathematical program library has been modified to include more thorough checking of the input data for violations of the program restrictions; to make use of one data point outside the range of integration in either direction, if such data are provided; and to use double precision arithmetic internally.

1.2 Applicable Programming Languages and Computer Systems

Except for DATA statements, the programming language is a common subset of Control Data 3600 FORTRAN, Control Data 6600 FORTRAN, and Univac 1108 FORTRAN V.

Applicable computing systems are the Control Data 3600--Tape SCOPE--and the Control Data 6600 SCOPE.

In each of the above computing systems, this routine is maintained for the convenience of the user in a library file. The routine is accessible by means of a few machine dependent control cards which are described in Appendices D and E.

1.3 Considerations Regarding Use

AVINT is meant to be used for integrating tabulated data only. Integration of functions described by formulae can be done more precisely with routines written for that purpose.[2]

2. Usage

2.1 Entry

CALL AVINT (X,Y,N,XLO,XUP,ANS,IERR)

2.2 Description of Arguments

X - Real array, dimensioned N, giving the abscissas at which functional values are supplied. These abscissa must be in increasing order. That is, it is required that X(I) < X(I+1) for I = 1,2,...,N-1.

Y - Real array, dimensioned N, giving the values of the function to be integrated. That is, Y(I) = f(X(I)) for I = 1,2,...,N.

N - An integer variable or constant telling how many functional values are stored in Y (= number of abscissas stored in X). N must be at least 3.

XLO - A real variable or constant giving the lower limit of integration.

XUP - A real variable or constant giving the upper limit of integration.

ANS - The real approximate value of the integral.

IERR - An integer error status parameter. IERR = 1 is normal. IERR ≥ 2 means the integral could not be evaluated. See Section 2.8.

2.3 Restrictions Between Arguments

XLO must be less than or equal to XUP, and there must be at least three X values between these limits, inclusive, if XLO ≠ XUP.

2.4 Principal Uses with Examples

The principal use of AVINT is to integrate tabulated data. This data may have been entered from some external medium such as cards or a tape. As an example, suppose x and f(x) are tabulated on 120 cards which are arranged in increasing value of x. Suppose the integral is to be evaluated from x = 0.0 to x = last value of x provided on cards. The following program segment suggests a way to do this integration:

```
      PROGRAM  INTEGR
      DIMENSION  X(120), Y(120)
      .
      .
      .
      READ 10, (X(I), Y(I), I=1, 120)
   10 FORMAT(---)
      XLO = 0.0
      XUP = X(120)
      CALL AVINT (X,Y,120,XLO,XUP,ANS,IERR)
      IF (IERR.NE.1) GO TO ---
      PRINT 20, ANS
   20 FORMAT(---)
      .
      .
      .
      END
```

2.5 Library Routines Explicitly Required

ERRCHK[2] and DBLE

2.6 User-Supplied Routines Required

None of the routines called by AVINT need be supplied by the user.

2.7 Cautions and Restrictions

The restrictions in AVINT are as follows: the X(I) must be strictly increasing, at least three X(I) must be between XLO and XUP unless XLO = XUP, and XLO must not be greater than XUP.

The user must not expect good answers when the data points are few and far between. Also XLO should not be made too much less than X(1), or XUP too much greater than X(N), because the extrapolation of a parabola far beyond the data points is bound to give an inaccurate estimate of the integral. Of course this problem can be precluded if the tabular data cover the range of integration, i.e., if X(1) ≤ XLO and XUP ≤ X(N).

If possible, data points should be picked more densely in regions where the function is changing more rapidly.

2.8 Error Conditions, Messages, and Codes

In the event that AVINT cannot estimate the desired integral because of some invalid input data, a call is made to the routine ERRCHK[2] to print an appropriate message, and IERR is set to the corresponding code:

If XLO > XUP, then IERR = 2 and the message is THE UPPER LIMIT OF INTEGRATION WAS NOT GREATER THAN THE LOWER LIMIT.

If N < 3 or if there are fewer than three X(I) between XLO and XUP, inclusive, then IERR = 3 and the message is THERE WERE LESS THAN THREE FUNCTION VALUES BETWEEN THE LIMITS OF INTEGRATION.

If X(I-1) ≥ X(I) for some I, then IERR = 4 and the message is THE ABSCISSA WERE NOT STRICTLY INCREASING. MUST HAVE X(I-1).LT.X(I) FOR ALL I.

These conditions are normally fatal errors and result in termination of execution. To make the conditions nonfatal, ERRSET[2] must be called before AVINT is called. For example, to make the errors nonfatal and set a maximum of 100 messages to be printed, the following call should be made:

```
CALL ERRSET(100,0)
```

The library error routine is described in detail in SC-M-69-337.

3. Mathematical Methods

3.1 Statement of Problem

We wish to approximate $\int_A^B F(x)\,dx$ where F is known only at a finite number of arbitrarily spaced points, say $F(x_i) = y_i$ for $i = 1,2,3,\ldots,N$. We restrict ourselves (without loss of generality) so that $A \le B$ and $x_{i-1} < x_i$ for $i = 2,3,\ldots,N$.

3.2 Methods Used

The method used is based on overlapping parabolas. This technique gives the effect of smoothing the data, since four data values are used in computing the integral between each pair of points (i.e., data at x_{i-1}, x_i, x_{i+1}, x_{i+2} are used in computing the integral from x_i to x_{i+1}).

3.3 Mathematical Range and Domain

This method is valid for any tabulated real function on a bounded interval.

3.4 Equations and Discussion

Any three successive data values of the form (x_{i-1}, y_{i-1}), (x_i, y_i), (x_{i+1}, y_{i+1}) define a unique parabola of the form $a_i x^2 + b_i x + c_i$ which passes through them. That parabola is determined from the following system of three equations in the three unknowns a_i, b_i, and c_i.

$$y_{i-1} = a_i x_{i-1}^2 + b_i x_{i-1} + c_i$$

$$y_i = a_i x_i^2 + b_i x_i + c_i$$

$$y_{i+1} = a_i x_{i+1}^2 + b_i x_{i+1} + c_i \quad .$$

Solving this system of equations gives

$$a_i = T_1 + T_2 + T_3$$

$$b_i = -(x_i + x_{i+1}) \cdot T_1 - (x_{i-1} + x_{i+1}) \cdot T_2 - (x_{i-1} + x_i) T_3$$

$$c_i = x_i \cdot x_{i+1} \cdot T_1 + x_{i-1} \cdot x_{i+1} \cdot T_2 + x_{i-1} \cdot x_i \cdot T_3 ,$$

where

$$T_1 = y_{i-1} / \left[(x_{i-1} - x_i) \cdot (x_{i-1} - x_{i+1}) \right]$$

$$T_2 = - y_i / \left[(x_{i-1} - x_i) \cdot (x_i - x_{i+1}) \right]$$

$$T_3 = y_{i+1} / \left[(x_{i-1} - x_{i+1}) \cdot (x_i - x_{i+1}) \right] .$$

Now let us assume that all the data values are inside the interval $A \leq x \leq B$ except possibly x_1 and x_N, for which we may have $x_1 < A$ or $x_N > B$. That is, if we have more than one data value that is less than A, then all these values are ignored except the one closest to A, and similarly for B. Then, we may define a_i, b_i, and c_i for all i such that $2 \leq i \leq N-1$. Now, for any i such that a_i, b_i, and c_i and a_{i+1}, b_{i+1}, and c_{i+1} are all defined (i.e., $2 \leq i \leq N-2$), we may approximate the integral from x_i to x_{i+1} as follows:

$$\int_{x_i}^{x_{i+1}} F(x)\,dx \approx 1/2 \int_{x_i}^{x_{i+1}} \left[\left(a_i x^2 + b_i x + c_i\right) + \left(a_{i+1}x^2 + b_{i+1}x + c_{i+1}\right)\right] dx$$

$$= \frac{a_i + a_{i+1}}{2}\left(\frac{x_{i+1}^3 - x_i^3}{3}\right) + \frac{b_i + b_{i+1}}{2}\left(\frac{x_{i+1}^2 - x_i^2}{2}\right)$$

$$+ \frac{c_i + c_{i+1}}{2} \cdot \left(x_{i+1} - x_i\right) .$$

The remaining problem is to evaluate the pieces of the integral which are left on each end. On the left end we use

$$\int_A^{x_2} F(x)\,dx \approx \int_A^{x_2} \left(a_2 x^2 + b_2 x + c_2\right) dx$$

$$= a_2\left(\frac{x_2^3 - A^3}{3}\right) + b_2\left(\frac{x_2^2 - A^2}{2}\right) + c_2\left(x_2 - A\right) ,$$

and on the right end

$$\int_{X_{N-1}}^{B} F(x)\,dx \approx \int_{X_{N-1}}^{B} \left(a_{N-1}x^2 + b_{N-1}x + c_{N-1}\right) dx$$

$$= a_{N-1}\left(\frac{B^3 - X_{N-1}^3}{3}\right) + b_{N-1}\left(\frac{B^2 - X_{N-1}^2}{2}\right)$$

$$+ c_{N-1}\left(B - X_{N-1}\right) .$$

Note that the last two integrals may or may not involve an "extrapolation" of the parabola beyond the data values that we used to define it. Extrapolation occurs if $A < X(1)$ or $X(N) < B$, but does not occur if $A \geq X(1)$ and $X(N) \geq B$.

3.5 Error Analysis, Bounds, and Estimates

Quantitative estimates of the error incurred when using the above described method are complicated by the fact that the abscissas used are arbitrarily spaced and the method involves an averaging. The following paragraphs show that under reasonable assumptions the method is about of order h^3, where h is the maximum spacing between abscissas, and the limits of integration lie within the range of abscissas given.

Let us consider the error incurred in integrating from x_i to x_{i+1} using only one parabola, say the one interpolating at x_{i-1}, x_i and x_{i+1}. Standard methods, as in Isaacson and Keller,[3] show that the error made in approximating the integral of the function F from x_i to x_{i+1} by the integral of the parabola is

$$E_i = \frac{F^{(3)}(\gamma)}{3!} \int_{x_i}^{x_{i+1}} \left(x - x_{i-1}\right)\left(x - x_i\right)\left(x - x_{i+1}\right) dx$$

for some $\gamma \in (c,d) = (x_{i-1}, x_{i+1})$, provided $F^{(3)}(x)$ is continuous on $[c,d]$. If we let $h_i = \max\left\{(x_i - x_{i-1}), (x_{i+1} - x_i)\right\}$, then

$$\begin{aligned} \left|E_i\right| &\leq \frac{1}{6} \max_{x\in(c,d)} \left|F^{(3)}(x)\right| \cdot \int_{x_i}^{x_{i+1}} 2h_i^3 \, dx \\ &= \frac{1}{6} \max_{x\in(c,d)} \left|F^{(3)}(x)\right| \cdot 2h_i^3 \cdot x \Big|_{x_i}^{x_{i+1}} \\ &\leq \frac{1}{6} \max_{x\in(c,d)} \left|F^{(3)}(x)\right| \cdot 2h_i^4 \\ &= 0\left(h_i^4\right). \end{aligned}$$

Now, the average of two order h_i^4 approximations would still be of order h_i^4. So, if we assume that the number of data points is like $\frac{B - A}{h}$, where h is the maximum interval between any two successive abscissas, and if we replace the h_i in the above analysis by h then we can conservatively say that the total error E incurred in AVINT's method is

$$E = O\left(\frac{1}{h}\right) \cdot O\left(h^4\right) = O\left(h^3\right).$$

Practically speaking it can be said that answers obtained using AVINT will generally be worse when the spacing of the x_i is coarser, and when the limits of integration are farther outside x_1 and x_N. Accuracy may be improved by providing one data value outside the range of integration in each direction, and by providing denser data values in areas where the curve is changing more rapidly.

4. Programming Methods

The coding is a straightforward implementation of the formulas and procedures described in the previous sections of this document. The specific modifications of the original program which were necessary include those listed in Section 1.1 as well as the inclusion of error messages printed through the routine ERRCHK[2] and the inclusion of the parameters ANS and IERR (AVINT was a FUNCTION originally). Double precision arithmetic is used internally. The contents of the X and Y arrays are converted to double precision and all internal calculations are then carried out in double precision. The final answer is then converted to single precision and stored in ANS. The double precision arithmetic is used because it was discovered during development of the routine (on the CDC 3600) that as N became larger, say $N > 100$, the use of double precision arithmetic often improved the accuracy by one or more significant figures. Also, it was felt that the somewhat increased execution time was not likely to be a significant factor in a routine intended for integration of tabular data.

5. Space, Time, and Accuracy Considerations

AVINT requires about 500_{10} core locations. The time required is approximately proportional to the number of function values given. Accuracy is dependent on the accuracy of the supplied data and the density of these values. See Appendices B and C.

6. Testing Methods

6.1 General

One way to test function integrator subroutines is to integrate functions whose integrals are evaluable in closed form. This kind of testing can be used to verify that a program is coded properly, provided the program is tested in all its operating modes. Testing of AVINT is complicated by the fact that nonequally spaced abscissas can be used, and that the ends of the interval of integration may be carried beyond the set of data points. So, if the amount of testing is to be kept at a reasonable level, certain sacrifices will be required. For example, if we fix our attention on a given function, a given interval, and a given number of points, we could integrate this function with arbitrarily many different distributions of the abscissas. Results of such tests would, at best, be difficult to correlate or relate to the user. Thus, it seems reasonable to restrict such tests to enough cases to determine that AVINT does give reasonable answers when the distribution of abscissas is reasonably even on the interval. This approach leaves it to the user to realize that AVINT is not magic, and the more extreme the distribution of abscissas is, the worse the answers are likely to be. Similar statements apply to the choice of what functions to integrate for test purposes, and how many different values of N (the number of abscissas) to use with a given type of distribution of abscissas, say equally spaced, to check for convergence of the integration process.

6.2 Kinds of Tests Used

To test AVINT various function/interval pairs were chosen and the integrals of these functions over their corresponding intervals were performed by AVINT using various numbers of function values (data points) and four different types of abscissa distributions: (1) equally spaced with the first and last data points on the ends of the interval of integration; (2) same as (1) except all abscissa perturbed by small random amounts; (3) equally spaced with all data values interior to the interval of integration; and (4) equally spaced with X(1) and X(N) exterior to the interval of integration.

6.3 Normal Cases Tested

Here are some examples of the integrals used:

$$\int_0^1 1\,dx, \quad \int_0^1 \frac{1}{1+x}\,dx, \quad \int_0^{2\pi} x \sin x \cos x\,dx, \text{ and } \int_0^1 \frac{x}{e^x - 1}\,dx.$$

6.4 Difficult Cases Tested

Here are some examples of the integrals used:

$$\int_{.01}^{1.1} x^{12}\,dx, \quad \int_{.01}^{1.1} \frac{1}{x}\,dx, \quad \int_{.01}^{1.1} \frac{1}{x^5}\,dx, \text{ and}$$

$$\int_0^{2\pi} x \sin(30x) \cos x\,dx\ .$$

6.5 Range, Error, and Fault Checks Tested

All the fault conditions (IERR = 2, 3, 4) were tested at least once by purposely submitting bad data.

7. Remarks

None required for this report.

8. Certification

This routine was subjected to a wide variety of tests. The performance of the routine throughout the tests was checked carefully. The nature of the tests, the reliability of the routine, the error analyses conducted, and the observed variation in accuracy are reported in this document. While it is believed that the facts recorded and the judgments expressed regarding accuracy and reliability are strong indications of the general quality and validity of the routine, the tests should not be considered to be exhaustive. The use of this routine outside of the stated range of application or in violation of stated restrictions may produce unspecified results. The statements made in this document are intended to apply only to those versions of the indicated routine which are released by the Sandia Laboratories Mathematical Program Library Project.

The author performed the modifications of AVINT listed in Section 1.1, performed the tests indicated in Section 6, and prepared this document. Dr. A. L. Roark originally suggested the use of this routine and provided advice on its testing. C. B. Bailey provided consultation, particularly regarding documentation.

APPENDIX A

The AVINT Listing

APPENDIX A

The AVINT Listing

```
      SUBROUTINE AVINT (X,Y,N,XLO,XUP,ANS,IERR)
C
C     SANDIA MATHEMATICAL PROGRAM LIBRARY
C     MATHEMATICAL COMPUTING SERVICES DIVISION 9422
C     SANDIA LABORATORIES
C     P. O. BOX 5800
C     ALBUQUERQUE, NEW MEXICO  87115
C
C     ORIGINAL PROGRAM FROM *NUMERICAL INTEGRATION* BY DAVIS+RABINOWITZ.
C     ADAPTATION AND MODIFICATIONS FOR SANDIA MATHEMATICAL PROGRAM
C     LIBRARY BY RONDALL E JONES.
C
C     CONTROL DATA 6600 VERSION
C
C     ABSTRACT
C         AVINT INTEGRATES TABULATED FUNCTIONS BY AN OVERLAPPING
C         PARABOLAS TECHNIQUE
C
C     DESCRIPTION OF PARAMETERS
C         X     - REAL ARRAY OF ABSCISSAS, WHICH MUST BE IN INCREASING
C                 ORDER.
C         Y     - REAL ARRAY OF FUNCTIONAL VALUES. I.E., Y(I)=FUNC(X(I))
C         N     - THE INTEGER NUMBER OF FUNCTION VALUES SUPPLIED. N.GE.3
C         XLO   - REAL LOWER LIMIT OF INTEGRATION
C         XUP   - REAL UPPER LIMIT OF INTEGRATION.  MUST HAVE XLO.LE.XUP.
C         ANS   - APPROXIMATE VALUE OF INTEGRAL
C         IERR  - INTEGER ERROR PARAMETER
C                 =1 IS NORMAL
C                 =2 MEANS XUP WAS LESS THAN XLO.
C                 =3 MEANS THE NUMBER OF X(I) BETWEEN XLO AND XUP
C                    (INCLUSIVE) WAS LESS THAN 3. NO INTEGRATION WAS DONE
C                 =4 MEANS THE RESTRICTION X(I+1).GT.X(I) WAS VIOLATED.
C                 ANS IS SET TO 0.0 IF IERR=2,3,OR 4.
C
C
      DIMENSION X(N),Y(N)
      DIMENSION MES2(8),MES3(9),MES4(9)
      DOUBLE PRECISION R3,RP5,SUM,SYL,SYL2,SYL3,SYU,SYU2,SYU3,X1,X2,X3
     *,X12,X13,X23,TERM1,TERM2,TERM3,A,B,C,CA,CB,CC
      DATA (MES2=2,69H THE UPPER LIMIT OF INTEGRATION WAS NOT GREATER TH
     *AN THE LOWER LIMIT.)
      DATA (MES3=3,78H THERE WERE LESS THAN THREE FUNCTION VALUES BETWEE
     *N THE LIMITS OF INTEGRATION.)
      DATA (MES4=4,80H THE ABSCISSA WERE NOT STRICTLY INCREASING.  MUST
     *HAVE X(I-1).LT.X(I) FOR ALL I.)
      IERR=1
      ANS =0.0
      IF (XLO-XUP) 3,100,200
    3 IF (N.LT.3) GO TO 205
      DO 5 I=2,N
    5 IF (X(I).LE.X(I-1))  GO TO 210
      IF (X(N-2).LT.XLO)   GO TO 205
      IF (X(3).GT.XUP)     GO TO 205
      I = 1
   10 IF (X(I).GE.XLO) GO TO 15
```

```
      I = I+1
      GO TO 10
   15 INLFT = I
      I = N
   20 IF (X(I).LE.XUP) GO TO 25
      I = I-1
      GO TO 20
   25 INRT = I
      IF ((INRT-INLFT).LT.2) GO TO 205
      ISTART = INLFT
      IF (INLFT.EQ.1) ISTART = 2
      ISTOP  = INRT
      IF (INRT.EQ.N)  ISTOP  = N-1
C
      R3 = 3.0D0
      RP5= 0.5D0
      SUM = 0.0
      SYL = XLO
      SYL2= SYL*SYL
      SYL3= SYL2*SYL
C
      DO 50 I=ISTART,ISTOP
      X1 = X(I-1)
      X2 = X(I)
      X3 = X(I+1)
      X12 = X1-X2
      X13 = X1-X3
      X23 = X2-X3
      TERM1 = DBLE(Y(I-1))/(X12*X13)
      TERM2 =-DBLE(Y(I)) /(X12*X23)
      TERM3 = DBLE(Y(I+1))/(X13*X23)
      A = TERM1+TERM2+TERM3
      B = -(X2+X3)*TERM1 - (X1+X3)*TERM2 - (X1+X2)*TERM3
      C = X2*X3*TERM1 + X1*X3*TERM2 + X1*X2*TERM3
      IF (I-ISTART) 30,30,35
   30 CA = A
      CB = B
      CC = C
      GO TO 40
   35 CA = 0.5*(A+CA)
      CB = 0.5*(B+CB)
      CC = 0.5*(C+CC)
   40 SYU = X2
      SYU2= SYU*SYU
      SYU3= SYU2*SYU
      SUM = SUM + CA*(SYU3-SYL3)/R3  + CB*RP5*(SYU2-SYL2) + CC*(SYU-SYL)
      CA  = A
      CB  = B
      CC  = C
      SYL = SYU
      SYL2= SYU2
      SYL3= SYU3
   50 CONTINUE
      SYU = XUP
```

```
      ANS = SUM + CA*(SYU**3-SYL3)/R3 + CB*RP5*(SYU**2-SYL2)              AVN 1090
     *   + CC*(SYU-SYL)                                                   AVN 1100
  100 RETURN                                                              AVN 1110
  200 IERR=2                                                              AVN 1120
      CALL ERRCHK(69,MES2)                                                AVN 1130
      RETURN                                                              AVN 1140
  205 IERR=3                                                              AVN 1150
      CALL ERRCHK(78,MES3)                                                AVN 1160
      RETURN                                                              AVN 1170
  210 IERR=4                                                              AVN 1180
      CALL ERRCHK(80,MES4)                                                AVN 1190
      RETURN                                                              AVN 1200
      END                                                                 AVN 1210
```

NOTE: The only change needed to run this routine on the CDC 3600 is to replace line AVN 360 by the following statement.

```
DIMENSION  MES2(10),MES3(11),MES4(11)
```

APPENDIX B

Test Results for CDC 3600

APPENDIX B

Test Results for CDC 3600

To test AVINT a set of fourteen test integrals, which included those listed in Sections 6.3 and 6.4, was chosen. Then, each integral was evaluated by AVINT for several different numbers and distributions of abscissas. The number of points used included, at various times, 10, 11, 40, 41, 52, 60, 100, 101, 200, 250, 500, 1000. Four types of distribution of abscissas were used. Specifically, they were as follows, where N is the number of points and A and B are the left and right limits of integration, respectively.

Type 1 -- Equally spaced with first and last data points on the ends of the interval. That is, the abscissas are A, $A + \Delta$, $A + 2\Delta$, ..., B, where

$$\Delta = (B - A)/(N - 1) .$$



Type 2 -- This distribution is the same as Type 1 except each abscissa is perturbed by a random amount between plus and minus $0.99 \cdot 1/2 \cdot \Delta$:



Type 3 -- Equally spaced with all data values interior to the interval of integration. That is, the abscissas are $A + 1/2\Delta$, $A + (1+1/2)\Delta$, $A + (2+1/2)\Delta$, ..., $A + (N-1/2)\Delta = B - 1/2\Delta$ where

$$\Delta = (B - A)/N.$$



Type 4 -- Equally spaced with first and last abscissas exterior to the interval of integration. That is, the abscissas are

$$A - 1/2\Delta,\ A + (1-1/2)\Delta,\ A + (2-1/2)\Delta,\ \ldots,\ A + (N-1-1/2)\Delta = B + 1/2\Delta,$$

where

$$\Delta = (B - A)/(N - 2).$$



Not all combinations of functions, numbers of abscissas, and distributions of abscissas listed above were performed, although a large portion of them were. In particular, all fourteen integrals were performed with each distribution type for N = 10, 40, 100, 250 and 1000. Sample tables of results are included for five of these functions with N = 10, 40, 100, and 250 and distribution Types 1, 3, and 4. The Type 2 distribution is not included because its results would not be very informative unless the actual lists of abscissas used were included, and this was deemed impractical. Generally, the answers for Type 2 distributions were about as accurate as the corresponding answer for Type 3 distributions--sometimes Type 2 was better, sometimes it was worse--but seldom was the error for a Type 2 more than a factor of ten either direction from the error in the corresponding Type 3 distribution.

Each table is divided into three segments for the answers for distribution Types 1, 3, and 4, in that order. The distribution type appears on the same line with EVALUATION OF AVINT. There are four columns of numbers. The first column may be ignored as it is just an identifier for the integral being performed. The second column gives the number of abscissas being used. The third column gives the answer computed by AVINT, and the fourth gives the relative error in this answer. The correct answer, which in each table is given at the bottom of the ANSWER column, was computed in double precision from the analytic expression for the integral whenever the analytic expression was available. The integral being evaluated is given at the top of each table.

Tables 1 and 2 show rather good results obtained for nontrivial but well-behaved functions. With extremely smooth functions the answers are better yet. For example, AVINT gives an answer correct to about 10 significant figures for

$$\int_0^1 \frac{1}{1+e^x}\,dx$$

when only 100 points are used with distribution Type 1, 3, or 4.

Table 3 which is for

$$\int_{.01}^{1.1} x^{12}\,dx$$

shows as good results as in Tables 1 and 2 after "enough" points have been used. For small numbers of points, the answers are not very good because of the small spike where the function rises rapidly from 1 at $x = 1.0$ to about 3.14 at $x = 1.1$.

Table 4 is for

$$\int_{.01}^{1.1} \frac{1}{x}\,dx \,,$$

which has a large spike at the left end of the interval where $\frac{1}{x}$ rises to 100. Note that for $N < 50$ the answers for a Type 4 distribution have to be bad because then $x(1)$ is negative. For a Type 1 distribution with 1000 points, the relative error in this integral is about 3.5×10^{-6}. Spikes of this type can be produced that are very bad indeed by using $f(x) = \frac{1}{x^n}$ for $n > 1$. For example, the relative error in the answer for

$$\int_{.01}^{1} \frac{1}{x^5}\,dx$$

using a Type 1 distribution with 100 points is about 0.8 (an 80% error!) and with 1000 points is about 1.7×10^{-3}. These relatively bad answers are of course due to the size of the spike in $\frac{1}{x^5}$ which rises to 10^{10} at $x = 0.01$. Hopefully experimental data will not be this wild.

Table 5 is for $\int_0^1 \sqrt{x}\,dx$. (For negative values of x, $-\sqrt{-x}$ was used for the integrand value.) The problem with this integral is the infinite (vertical) slope at x = 0. This problem is not nearly so bad as the spikes in the above cases, but the infinite slope does degrade the answers somewhat since a parabola cannot have an infinite slope, and AVINT uses parabolas to fit the data.

Other types of difficulties can occur: for example, oscillation. In evaluating

$$\int_0^{2\pi} x \sin (30x) \cos x \, dx$$

using a Type 1 distribution, AVINT had these relative errors:

40 points	-4.47
250 points	6.75×10^{-3}
1000 points	3.29×10^{-5} .

Two problems team up to cause this lack of precision: More data points are required to describe a highly oscillating function; also there is much cancellation of areas above the X-axis with similar areas below the X-axis, which causes a loss of precision.

Table 6 is similar to Table 5 except odd numbers of points have been used, and N ≈ 250 has not been used. Note that these results agree very nearly with Table 5. This demonstrates that AVINT is not particularly sensitive to small changes in the number of points.

Note that in all the tables the answers for the Type 4 distribution are the best. This is to be expected since interpolation between data points, when there are data values outside the interval of integration, is generally more accurate than extrapolation outside the range of data values as in a Type 3 distribution.

In summary, AVINT appears to perform as desired and the accuracy of its answers appears to depend mainly on the smoothness of behavior of the function and the density of data values. Also it helps, in general, if one data value is available outside the interval of integration in each direction.

TABLE 1

Results for $\int_0^{2\pi} x \sin x \cos x \, dx$

FUNCTION	NUM FUNC EVAL	ANSWER	REL ERR
EVALUATION OF AVINT 1			
14	10	-1,5695301303+000	-8,0608572066-004
14	40	-1,5711556739+000	2,2876749814-004
14	100	-1,5708063397+000	6,3744339759-005
14	250	-1,5707965886+000	1,6665839304-007
EVALUATION OF AVINT 3			
14	10	-1,6018708585+000	1,9782661294-002
14	40	-1,5703646956+000	-2,7478493748-004
14	100	-1,5707826192+000	-8,7265576958-006
14	250	-1,5707959525+000	-2,3825413442-007
EVALUATION OF AVINT 4			
14	10	-1,5993230456+000	1,8160673330-002
14	40	-1,5708517918+000	3,5310116106-005
14	100	-1,5707975602+000	7,9794692766-007
14	250	-1,5707963573+000	1,9434317667-008
		-1,5707963268+000 *	

TABLE 2

$$\text{Results for } \int_0^4 e^x \, dx$$

EVALUATION OF AVINT 1

FUNCTION	NUM FUNC EVAL	ANSWER	REL ERR
21	10	5,3636694208+001	7,1913257362-004
21	40	5,3598293986+001	2,6857835047-006
21	100	5,3598153680+001	6,8033829697-008
21	250	5,3598150125+001	1,7269339815-009

EVALUATION OF AVINT 3

21	10	5,3563995138+001	-6,3724019043-004
21	40	5,3597964360+001	-3,4641636967-006
21	100	5,3598144952+001	-9,4796862060-008
21	250	5,3598149900+001	-2,4954390947-009

EVALUATION OF AVINT 4

21	10	5,3608028387+001	1,8430401983-004
21	40	5,3598169453+001	3,6233149550-007
21	100	5,3598150471+001	8,1734377316-009
21	250	5,3598150044+001	1,9784414321-010
		5,3598150033+001 *	

TABLE 3

Results for $\int_{.01}^{1.1} x^{12}\,dx$

EVALUATION OF AVINT 1

FUNCTION	NUM FUNC EVAL	ANSWER	REL ERR
2	10	2,7381004620-001	3,1069223553-002
2	40	2,6560120179-001	1,5769586176-004
2	100	2,6556044846-001	4,2336177251-006
2	250	2,6555935352-001	1,1048238862-007

EVALUATION OF AVINT 3

FUNCTION	NUM FUNC EVAL	ANSWER	REL ERR
2	10	2,5889506559-001	-2,5095178310-002
2	40	2,6550699897-001	-1,9703775691-004
2	100	2,6555777864-001	-5,8199477720-006
2	250	2,6555928229-001	-1,5775004970-007

EVALUATION OF AVINT 4

FUNCTION	NUM FUNC EVAL	ANSWER	REL ERR
2	10	2,6871357872-001	1,1877777388-002
2	40	2,6556554192-001	2,3413743988-005
2	100	2,6555946476-001	5,2937980939-007
2	250	2,6555932763-001	1,2970718755-008
		2,6555932418-001 *	

TABLE 4

$$\text{Results for} \int_{.01}^{1.1} \frac{1}{x}\,dx$$

EVALUATION OF AVINT 1

FUNCTION	NUM FUNC EVAL	ANSWER	REL ERR
3	10	7,3309630003+000	5,5961996005-001
3	40	4,9314076249+000	4,9128438178-002
3	100	4,7282947423+000	5,9173476617-003
3	250	4,7024676324+000	4,2277946433-004

EVALUATION OF AVINT 3

FUNCTION	NUM FUNC EVAL	ANSWER	REL ERR
3	10	3,9611139930+000	-1,5729591770-001
3	40	4,5619425777+000	-2,9473112809-002
3	100	4,6772391264+000	-4,9444392008-003
3	250	4,6983729920+000	-4,4833158315-004

EVALUATION OF AVINT 4

FUNCTION	NUM FUNC EVAL	ANSWER	REL ERR
3	10	3,5592770574+000	-2,4278440065-001
3	40	4,1805251987+000	-1,1061745325-001
3	100	4,7310503270+000	6,5035823539-003
3	250	4,7011375683+000	1,3981603845-004
		4,7004803658+000 *	

TABLE 5

Results for $\int_0^1 \sqrt{x}\ dx$

EVALUATION OF AVINT 1

FUNCTION	NUM FUNC EVAL	ANSWER	REL ERR
8	10	6,6295950039-001	-5,5607494141-003
8	40	6,6625549030-001	-6,1676454789-004
8	100	6,6656500048-001	-1,5249929129-004
8	250	6,6664117896-001	-3,8231555664-005

EVALUATION OF AVINT 3

8	10	6,6769567251-001	1,5435087698-003
8	40	6,6679553015-001	1,9329522911-004
8	100	6,6669926821-001	4,8902322304-005
8	250	6,6667491430-001	1,2371456250-005

EVALUATION OF AVINT 4

8	10	6,6675368551-001	1,3052826398-004
8	40	6,6667504678-001	1,2570177205-005
8	100	6,6666868997-001	3,0349547160-006
8	250	6,6666716926-001	7,5389834819-007
		6,6666666667-001 *	

TABLE 6

Results for $\int_0^1 \sqrt{x}\ dx$

EVALUATION OF AVINT 1

FUNCTION	NUM FUNC EVAL	ANSWER	REL ERR
8	11	6,6350102284-001	-4,7484657479-003
8	41	6,6627081238-001	-5,9378143487-004
8	101	6,6656652164-001	-1,5021753643-004

EVALUATION OF AVINT 3

FUNCTION	NUM FUNC EVAL	ANSWER	REL ERR
8	11	6,6755898963-001	1,3384844497-003
8	41	6,6679084486-001	1,8626728706-004
8	101	6,6669878523-001	4,8177855205-005

EVALUATION OF AVINT 4

FUNCTION	NUM FUNC EVAL	ANSWER	REL ERR
8	11	6,6673953530-001	1,0930294957-004
8	41	6,6667472651-001	1,2089767551-005
8	101	6,6666865939-001	2,9890943551-006
		6,6666666667-001 *	

APPENDIX C

Test Results for CDC 6600

APPENDIX C

Test Results for CDC 6600

To test AVINT a set of fourteen test integrals, which included those listed in Sections 6.3 and 6.4, was chosen. Then, each integral was evaluated by AVINT for several different numbers and distributions of abscissas. The number of points used included, at various times, 10, 40, 41, 100, 250, 251, and 1000. Four types of distribution of abscissas were used. Specifically, they were as follows, where N is the number of points and A and B are the left and right limits of integration, respectively.

<u>Type 1</u> -- Equally spaced with first and last data points on the ends of the interval. That is, the abscissas are A, A + Δ, A + 2Δ, ..., B, where

$$\Delta = (B - A)/(N - 1).$$



<u>Type 2</u> -- This distribution is the same as Type 1 except each abscissa is perturbed by a random amount between plus and minus 0.99 · 1/2 · Δ:



<u>Type 3</u> -- Equally spaced with all data values interior to the interval of integration. That is, the abscissas are A + 1/2Δ, A + (1+1/2)Δ, A + (2+1/2)Δ, ..., A + (N-1/2)Δ = B - 1/2Δ where

$$\Delta = (B - A)/N.$$



Type 4 -- Equally spaced with first and last abscissas exterior to the interval of integration. That is, the abscissas are A - 1/2Δ A + (1-1/2)Δ, A + (2-1/2)Δ, ..., A + (N-1-1/2)Δ = B + 1/2Δ, where

$$\Delta = (B - A)/(N - 2) .$$



All fourteen integrals were performed with each distribution type for N = 10, 40, 100, and 250, and distributions 1, 3, and 4 were also performed with N = 41 and 251. Sample tables of results are included for five of these functions with N = 10, 41, 100, 251, and 1000, for distribution Types 1, 3, and 4. The Type 2 distribution is not included because its results would not be very informative unless the lists of abscissas actually used were included, and this was deemed impractical. Generally, the answers for Type 2 distributions were about as accurate as the corresponding answer for Type 3 distributions--often Type 2 was better, sometimes it was worse--but seldom was the error for a Type 2 more than a factor of ten either direction from the error in the corresponding Type 3 distribution.

Each table is divided into three segments for the results of distribution Types 1, 3, and 4, in that order. The distribution type appears on the same line with EVALUATION OF AVINT. There are four columns of numbers. The first column may be ignored as it is just an identifier for the integral being performed. The second column gives the number of abscissas being used. The third column gives the answer computed by AVINT, and the fourth gives the relative error in this answer. The correct answer, which in each table is given at the bottom of the ANSWER column, was computed in double precision from the analytic expression for the integral, whenever the analytic expression was available as it was for eleven of the fourteen integrals used. The integral being evaluated is given at the top of each table.

Tables 1 and 2 show rather good results obtained for nontrivial but well-behaved functions. With extremely smooth functions the answers are better yet. For example, AVINT gives an answer correct to about 10 significant figures for

$$\int_0^1 \frac{1}{1+e^x} \, dx$$

when only 100 points are used with distribution Type 1, 3, or 4.

Table 3, which is for

$$\int_{.01}^{1.1} x^{12} \, dx$$

shows very good results after "enough" points have been used. For small numbers of points, the answers are not very good because of the small spike where the function rises rapidly from 1 at $x = 1.0$ to about 3.14 at $x = 1.1$.

Table 4 is for

$$\int_{.01}^{1.1} \frac{1}{x} \, dx,$$

which has a large spike at the left end of the interval where $\frac{1}{x}$ rises to 100. Note that for $N < 50$ the answers for a Type 4 distribution have to be bad because then Y(1) is large and negative. Taking this difficulty into account, the answers seem quite acceptable. Spikes of this type can be produced that are very bad indeed by using $f(x) = \frac{1}{x^n}$ for $n > 1$. For example, the relative error in the answer for

$$\int_{.01}^{1} \frac{1}{x^5} \, dx$$

using a Type 1 distribution with 100 points is about 0.8 (an 80% error!) and with 1000 points is about 1.7×10^{-3}. These relatively bad answers are, of course, due to the size of the spike in $\frac{1}{x^5}$, which rises to 10^{10} at $x = 0.01$. Hopefully experimental data will not be this wild.

Table 5 is for $\int_0^1 \sqrt{x}\, dx$. (For negative values of x, $-\sqrt{-x}$ was used for the integrand value.) The problem with this integral is the infinite (vertical) slope at x = 0. This problem is not nearly so bad as the spikes in the above cases, but the infinite slope does degrade the answers somewhat since a parabola cannot have an infinite slope, and AVINT uses parabolas to fit the data.

Other types of difficulties can occur: for example, oscillation. In evaluating

$$\int_0^{2\pi} x \sin(30x) \cos x \, dx$$

using a Type 1 distribution, AVINT had these relative errors:

41 points	-4.12
251 points	6.66×10^{-3}
1000 points	3.29×10^{-5} .

Two problems team up to cause this lack of precision: More data points are required to describe a highly oscillating function; also there is much cancellation of areas above the X-axis with similar areas below the X-axis, which causes a loss of precision.

Note that the answers for the Type 4 distribution are consistently the best. This is to be expected since interpolation between data points, when there are data values outside the interval of integration, is generally more accurate than extrapolation outside the range of data values as in a Type 3 distribution.

It should be pointed out that for distributions of Types 1, 3, and 4, AVINT's answers almost always improved very smoothly as the number of function values was increased. This was observed when comparing results for 40 and 41 function values, and for 250 and 251. Because of this there need be no attempt to make the number of function values used a nice round number, or an even number.

In summary, AVINT appears to perform as desired and the accuracy of its answers appears to depend mainly on the smoothness of behavior of the function and the density of data values. Also it helps, in general, if one data value is available outside the interval of integration in each direction.

TABLE 1

Results for $\int_0^{2\pi} x \sin x \cos x \, dx$

EVALUATION OF AVINT 1

FUNCTION	NUM FUNC EVAL	ANSWER	REL ERR
14	10	-1.5695301304968E+00	-8.0608559909782E-04
14	41	-1.5711233261897E+00	2.0817428028362E-04
14	100	-1.5708063397632E+00	6.3744535985936E-06
14	251	-1.5707965844855E+00	1.6405094660431E-07
14	1000	-1.5707963278261E+00	6.5646305468069E-10
EVALUATION OF AVINT		3	
14	10	-1.6018708584432E+00	1.9782661264344E-02
14	41	-1.5704011601325E+00	-2.5157091065903E-04
14	100	-1.5707826192016E+00	-8.7265249278022E-06
14	251	-1.5707959584572E+00	-2.3449105365452E-07
14	1000	-1.5707963252916E+00	-9.5703092164589E-10
EVALUATION OF AVINT		4	
14	10	-1.5993230457506E+00	1.8160673327948E-02
14	41	-1.5708463172049E+00	3.1824883461850E-05
14	100	-1.5707975802434E+00	7.9797011036356E-07
14	251	-1.5707963568689E+00	1.9145734773389E-08
14	1000	-1.5707963269114E+00	7.4176715941388E-11
		-1.5707963267949D+00 *	

TABLE 2

Results for $\int_0^4 e^x \, dx$

EVALUATION OF AVINT 1

FUNCTION	NUM FUNC EVAL	ANSWER	REL ERR
21	10	5.3636694208336E+01	7.1913256648447E-04
21	41	5.3598280393854E+01	2.4321867456005E-06
21	100	5.3598153680045E+01	6.8041537759845E-08
21	251	5.3598150124632E+01	1.7069235589928E-09
21	1000	5.3598150033506E+01	6.7588197357206E-12

EVALUATION OF AVINT 3

FUNCTION	NUM FUNC EVAL	ANSWER	REL ERR
21	10	5.3563995138489E+01	-6.3724017777447E-04
21	41	5.3597981368383E+01	-3.1468392355571E-06
21	100	5.3598144952580E+01	-9.4789909106118E-08
21	251	5.3598149901665E+01	-2.4530511062072E-09
21	1000	5.3598150032615E+01	-9.8705752562017E-12

EVALUATION OF AVINT 4

FUNCTION	NUM FUNC EVAL	ANSWER	REL ERR
21	10	5.3608028387973E+01	1.8430402583785E-04
21	41	5.3598167537325E+01	3.2658180880451E-07
21	100	5.3598150472189E+01	8.1914246281223E-09
21	251	5.3598150043679E+01	1.9654603227363E-10
21	1000	5.3598150033185E+01	7.6035939934860E-13

5.3598150033144D+01 *

TABLE 3

Results for $\int_{.01}^{1.1} x^{12}\,dx$

FUNCTION	NUM FUNC EVAL	ANSWER	REL ERR
EVALUATION OF AVINT		1	
2	10	2.7381004618192E-01	3.1069223508469E-02
2	41	2.6559733782307E-01	1.4314556307229E-04
2	100	2.6556044845117E-01	4.2335816574293E-06
2	251	2.6555935304124E-01	1.0866555142310E-07
2	1000	2.6555932429965E-01	4.3518359906223E-10
EVALUATION OF AVINT		3	
2	10	2.5889506559522E-01	-2.5095178297119E-02
2	41	2.6551164445534E-01	-1.7954454767736E-04
2	100	2.6555777863857E-01	-5.8199632707376E-06
2	251	2.6555928293585E-01	-1.5532587431202E-07
2	1000	2.6555932401559E-01	-6.3447945802714E-10
EVALUATION OF AVINT		4	
2	10	2.6871357866966E-01	1.1877777198258E-02
2	41	2.6556492835779E-01	2.1103283491021E-05
2	100	2.6555946476455E-01	5.2937498653986E-07
2	251	2.6555932755726E-01	1.2702150790215E-08
2	1000	2.6555932419714E-01	4.9148087510899E-11
		2.6555932418408D-01 *	

TABLE 4

$$\text{Results for } \int_{.01}^{1.1} \frac{1}{x}\,dx$$

EVALUATION OF AVINT 1

FUNCTION	NUM FUNC EVAL	ANSWER	REL ERR
3	10	7.3309630093989E+00	5.5961996198299E-01
3	41	4.9200076388261E+00	4.6703157113745E-02
3	100	4.7282947424112E+00	5.9173476866648E-03
3	251	4.7024427737682E+00	4.1749094199998E-04
3	1000	4.7004966734635E+00	3.4693626589268E-06

EVALUATION OF AVINT 3

FUNCTION	NUM FUNC EVAL	ANSWER	REL ERR
3	10	3.9611139932698E+00	-1.5729591764776E-01
3	41	4.5673983162727E+00	-2.8312435998706E-02
3	100	4.6772391267806E+00	-4.9444391217768E-03
3	251	4.6983975536679E+00	-4.4310622797489E-04
3	1000	4.7004589571086E+00	-4.5545736146557E-06

EVALUATION OF AVINT 4

FUNCTION	NUM FUNC EVAL	ANSWER	REL ERR
3	10	3.5592770578987E+00	-2.4278440054740E-01
3	41	4.1701500381691E+00	-1.1282470861547E-01
3	100	4.7310503267561E+00	6.5035823117545E-03
3	251	4.7011270681099E+00	1.3758217610559E-04
3	1000	4.7004828835562E+00	5.3563968088109E-07

4.7004803657924D+00 *

TABLE 5

Results for $\int_0^1 \sqrt{x}\, dx$

EVALUATION OF AVINT 1

FUNCTION	NUM FUNC EVAL	ANSWER	REL ERR
8	10	6.6295950038642E-01	-5.5607494203631E-03
8	41	6.6627081237294E-01	-5.9378144059607E-04
8	100	6.6656500047274E-01	-1.5249929088945E-04
8	251	6.6664133173430E-01	-3.8002398554226E-05
8	1000	6.6666349504357E-01	-4.7574346524470E-06

EVALUATION OF AVINT 3

8	10	6.6769567250978E-01	1.5435087646747E-03
8	41	6.6679084485866E-01	1.8626728799020E-04
8	100	6.6669926822383E-01	4.8902335745282E-05
8	251	6.6667486507515E-01	1.2297612721923E-05
8	1000	6.6666769762313E-01	1.5464346887484E-06

EVALUATION OF AVINT 4

8	10	6.6675368551465E-01	1.3052827197413E-04
8	41	6.6667472652020E-01	1.2089780298652E-05
8	100	6.6666868997348E-01	3.0349602155866E-06
8	251	6.6666716624030E-01	7.4936045280083E-07
8	1000	6.6666672892571E-01	9.3388567989905E-08

6.6666666666667D-01 *

APPENDIX D

Control Cards for Using AVINT on the CDC 3600

APPENDIX D

Control Cards for Using AVINT on the CDC 3600

AVINT is maintained on an auxiliary library tape for the convenience of the Control Data 3600 users at Sandia Laboratories, Albuquerque, New Mexico. The tape is labeled 36-00001 and is in HI (556 BPI) density. Questions concerning the availability of AVINT on the Control Data 3600 at Sandia Laboratories, Livermore, California, should be directed to the Numerical Applications Division 8321.

Two control cards, EQUIP and LIBRARY, are required for using the auxiliary library tape. The EQUIP card may immediately precede the LIBRARY card or may appear at the beginning of the job. The LIBRARY card must precede the first binary deck or first LOAD card (for an execution in which the auxiliary library is needed). If the job includes a compilation, then the LIBRARY card should appear between the SCOPE and LOAD cards.

A complete typical example follows:

${}^{7}_{9}$JOB,...

${}^{7}_{9}$FTN,L,X

⋮

SCOPE

${}^{7}_{9}$EQUIP,72=(36-00001),HI,RO

${}^{7}_{9}$LIBRARY,72

${}^{7}_{9}$LOAD

${}^{7}_{9}$RUN

All library routines required by a program must be available on a single library tape. Auxiliary library tape 36-00001 contains routines which are likely to be used in connection with the mathematical library routines. In particular, the tape includes the standard Control Data 3600 FORTRAN routines (as modified by Sandia), the SCORS SC4020 plot package, and a few other special Sandia routines (GOFU, ROMBERG, DATE, ANDGEN, and UDGEN).

APPENDIX E

Control Cards for Using AVINT on the CDC 6600

APPENDIX E

Control Cards for Using AVINT on the CDC 6600

AVINT is maintained in a library file for the convenience of the Control Data 6600 users at Sandia Laboratories, Albuquerque, New Mexico. The name of the file is MATHLIB. Questions concerning the availability of AVINT on the Control Data 6600 at Sandia Laboratories, Livermore, California, should be directed to the Numerical Applications Division 8321.

One control card, COLLECT, is required for using the mathematical library file. The COLLECT processor operates on one relocatable binary file and from one to six library files. The library files are searched for routines which contain entry points matching external references in the relocatable binary file. Such routines are added to the relocatable binary file.

A complete typical example follows:

```
JOB CARD
ACCOUNT CARD
FUN,S.
COLLECT,LGO,MATHLIB.
REDUCE.
LGO.
7/8/9 punch in column 1
Program
7/8/9 punch in column 1
Data
6/7/8/9 punch in column 1
```

In the above example, external references in LGO are satisfied, if possible, by selectively adding routines to LGO from MATHLIB. Additional information on the COLLECT processor with examples is contained in UR0004/6600.[4]

References

1. Philip J. Davis and Philip Rabinowitz, Numerical Integration, Blaisdell Publishing Company, 1967.

2. C. B. Bailey, Mathematical Program Library, SC-M-69-337, Sandia Laboratories (to be published soon; in addition to the description of ERRCHK and ERRSET, includes an index to other routines in the library).

3. Eugene Isaacson and H. B. Keller, Analysis of Numerical Methods, John Wiley & Sons, 1966, pp. 303 ff.

4. P. A. Lemke, Auxiliary Library Routines (PREP and COLLECT), Sandia Computing Publication UR0004/6600, June 1969.

DISTRIBUTION:

D. R. Cotter, 100
W. C. Myre, 1210
O. M. Stuetzer, 1220
D. M. Olson, 1510
V. E. Blake, Jr., 1520
T. B. Lane, 1540
R. T. Othmer, 1541
R. L. Peurifoy, Jr., 1550
I. M. Moore, 1610
B. E. Arthur, Jr., 1640
R. R. Prairie, 1643
W. L. Stevens, 1650
C. Winter, 1710
M. J. Norris, 1720
H. H. Wicke, 1721
R. J. Thompson, 1722
D. R. Morrison, 1724
R. G. Clem, 1730
H. D. Sivinski, 1740
A. A. Lieber, 1910
C. B. McCampbell, 2310
K. L. Gillespie, 2320
J. P. Shoup, 2340
J. W. Jones, 2350
W. C. Kraft, 2440
L. J. Paddison, 2450, Actg.
G. H. Roth, 2490
T. S. Church, 2610
G. W. Rodgers, 2620
E. G. Franzak, 2630
J. R. Sublett, 2640
G. W. Gobeli, 5110
O. E. Jones, 5130
J. E. Schirber, 5150
W. Herrmann, 5160
L. D. Bertholf, 5162
C. H. Karnes, 5165
A. W. Snyder, 5220
C. R. Mehl, 5230
J. H. Renken, 5231
G. W. McClure, 5232
E. H. Beckner, 5240
F. W. Neilson, 5260
J. E. McDonald, 5310, Actg.
R. L. Schwoebel, 5330
W. M. O'Neill, 5520
L. M. Berry, 5530
R. L. Brin, 7250
T. L. Pace, 7260
J. E. Stiegler, 7280
W. T. Moffat, 7290
M. McWhirter, 7320
M. Cowan, 7340
F. Neilson, 7360
S. A. Moore, 7370
J. R. Meikle, 7410
R. H. Schultz, 7420
A. P. Gruer, 7430
W. E. Boyes, 7450
L. E. Lamkin, 7510, Actg.
W. E. Caldes, 7520
R. J. Dill, 7530
J. W. Pearce, 8120
J. E. Benton, 8121
D. E. Gregson, 8130
L. E. Davies, 8150
R. A. Baroody, 8160
G. E. Brandvold, 8170
C. S. Selvage, 8180
A. N. Blackwell, 8310
J. L. Wirth, 8320
J. N. Rogers, 8321
A. G. Schuknecht, 8322
G. W. Anderson, Jr., 8330
C. D. Broyles, 9110
G. E. Hansche, 9120
H. E. Viney, 9130
J. R. Banister, 9150
M. L. Kramm, 9210
J. C. Eckhart, 9220
H. H. Patterson, 9230
R. E. Hepplewhite, 9240
T. F. Marker, 9250
R. C. Maydew, 9320
W. H. Curry, 9322
E. C. Rightley, 9323
H. R. Vaughn, 9325
D. F. McVey, 9328
K. J. Touryan, 9340
W. C. Scrivner, 9400
L. E. Mahuron, 9410
J. E. Wesbrook, 9411
C. E. Katzenberger, 9412
R. Lynes, 9413
G. D. Horne, Jr., 9414
M. B. Moore, 9416
J. L. Tischhauser, 9420
D. A. Young, 9421
A. R. Iacoletti, 9422
R. E. Jones, 9422 (50)
A. J. Arenholz, 9423
D. K. Robbins, 9424
E. K. Montoya, 9425
E. G. Thuman, 9426
B. T. Fox, 9427
P. Van Delinder, 9427
For: COSMIC (3)
A. J. Clark, Jr., 9510
G. J. Hildebrandt, 9520
M. S. Goldstein, 3416 (2)
L. C. Baldwin, 3412
J. L. Gardner, 3421
C. H. Sproul, 3428-2 (65)
B. F. Hefley, 8232 (5)